Solve $y' + 3y = \sin x$ with $y(0) = 1$ with *Mathematica*

Define p and q, and compute the integrating factor

To put the equation in standard form $y' + p(x)\,y = q(x)$ we set $p(x) = 3$and $q(x) = \sin x$.

```
p[x_] = 3
```

3

```
q[x_] = Sin[x]
```

$\sin(x)$

The integrating factor is $\mu = e^{\int p(x)\,\mathrm{dx}}$.

```
mu[x_] = Exp[Integrate[p[x], x]]
```

e^{3x}

Define values for the initial conditions x_0 and y_0

```
x0 = 0
```

0

```
y0 = 1
```

1

Use the standard solution formula to compute $y(x)$

The solution is
$y(x) = \frac{1}{\mu(x)}\left[\mu(x_0)\,y(x_0) + \int_{x_0}^{x}\mu(x)\,q(x)\,dx\right]$
which is entered into *Mathematica* as

```
y[x_] = 1/mu[x] (mu[x0] y0 + Integrate[mu[x] q[x], {x, x0, x}])
```

$$e^{-3x}\left(\frac{1}{10}\left(1 - e^{3x}(\cos(x) - 3\sin(x))\right) + 1\right)$$

Running Expand[ ] multiplies everything out

```
Expand[y[x]]
```

$$-\frac{\cos(x)}{10} + \frac{11\,e^{-3x}}{10} + \frac{3\sin(x)}{10}$$

This is the same answer obtained by hand previously.

Check the solution

Plug $y(x)$ into the LHS, and compare to the RHS.

```
LHS = FullSimplify[D[y[x], x] + p[x] y[x]]
```

$\sin(x)$

```
RHS = q[x]
```

$\sin(x)$

These are the same,so LHS=RHS and the equation checks

```
LHS == RHS
```

True

Verify the solution satisfies the initial conditions

```
y[x0] == y0
```

True

All done!